# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=pantheon-calculator
pkgver=2.0.2
pkgrel=1
pkgdesc='The Pantheon Calculator'
arch=(loong64 x86_64)
url=https://github.com/elementary/calculator
license=(GPL3)
groups=(pantheon)
depends=(
  glib2
  libgranite-7.so
)
makedepends=(
  git
  intltool
  meson
  vala
)
_tag=6d19987a92611645b7849565f657593344d4bb2f
source=(pantheon-calculator::git+https://github.com/elementary/calculator.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd pantheon-calculator
  git describe --tags
}

build() {
  arch-meson pantheon-calculator build
  ninja -C build
}

package() {
  DESTDIR="${pkgdir}" ninja -C build install
}

# vim: ts=2 sw=2 et:
